

# 标签打印机

---

## 中文编程手册

### 文件版本说明

日期	版本	修订内容
2018	V1.0	初稿

系统设定指令.....	1
SIZE.....	1
GAP.....	2
OFFSET.....	3
SPEED.....	4
DENSITY.....	5
DIRECTION.....	6
REFERENCE.....	7
CODEPAGE.....	8
CLS.....	9
FEED.....	10
BACKFEED&BACKUP.....	11
HOME.....	13
PRINT.....	14
SOUND.....	15
SELFTEST.....	16
卷标内容设计指令.....	17
BARCODE.....	18
BITMAP.....	19
BOX.....	21
PUTBMP.....	23
PUTPCX.....	23
QRCODE.....	24
TEXT.....	26
询问打印机状态指令.....	27
<ESC>!?...?	28
~!T.....	29
文件管理指令.....	30
DOWNLOAD.....	30
EOP.....	32
FILES.....	33
打印机外围功能设定指令.....	34
SET COUNTER.....	34
SET PEEL.....	35
SET TEAR & SET STRIPPER.....	36
BEEP.....	37

# 系统设定指令

## SIZE

功能：定义标签纸的宽度及高度

语法：(1) 英制系统 (英寸)

SIZE m,n

(2) 公制系统 (毫米)

SIZE m mm, n mm

参 数	说 明
m	定义标签纸的宽度 (英寸或毫米)
n	定义标签纸的高度 (英寸或毫米)

注意事项：

**200 DPI: 1 mm = 8 dots**

**300 DPI: 1mm = 12 dots**

在公制系统下, 在参数 *m* 与 “mm” 之间必须要有空格分隔


范例：

(1) 英制系统 (英寸)

SIZE 3.5, 3.00

(2) 公制系统 (毫米)

SIZE 100 mm, 100 mm



其他参考项目：GAP

## GAP

功能：定义两张标签之间的距

语法：(1) 英制系统 (英寸)

GAP m,n

(2) 公制系统(毫米)

GAP m mm, n mm

参 数	说 明
m	定义标签间隙高度 (英寸或毫米) $0 \leq m \leq 1$ (inch), $0 \leq m \leq 25.4$ (mm)
n	定义标签间隙高度的补偿值 (英寸或毫米) $n \leq \text{label length}$ (inch or mm)
0,0	表示连续纸模式

注意事项：

在公制系统下,在参数 *m* 与 “mm” 之间必须要有空格分隔

范例：一般标签纸

(1) 英制系统 (inch)


GAP 0.12,0

(2) 公制系统 (mm)

GAP 3 mm,0

(3) 连续纸模式

GAP 0,0



其他参考项目：

SIZE

## OFFSET

功能:

定义标签于打印完后额外推出的长度，特别是在使用自动剥纸器或裁刀功能时，可用来调校标签停止的位置，在打印下一张时打印机会将原先多推出或少推出的部分以回拉方式补偿回来。此方式适用于剥离模式和裁刀模式。

语法:

(1) 英制系统 (英寸)

OFFSET m

(2) 公制系统 (毫米)

OFFSET m mm

参数	说明
m	偏移距离 (英寸或毫米) $-1 \leq m \leq 1(\text{inch})$

警告: 不适当的偏移值会导致出现“卡纸”的现象


范例:

(1) 英制系统 (英寸)

OFFSET 0.5

(2) 公制系统 (毫米)

OFFSET 12.7 mm



其他参考项目:

SIZE, GAP, SET PEEL

## SPEED

功能：设定打印机的打印速度

语法：SPEED n

<u>参</u>	<u>数</u>
n	

<u>说</u>	<u>明</u>
每秒的打印速度，以英寸计算	

范例：

SPEED 4

其他参考项目：

DENSITY

---

## DENSITY

功能：设定打印机的打印浓度

语法：DENSITY n

参 数
n

说 明
1~7
1 表示最淡的浓度
7 表示最深的浓度

范例：

DENSITY 3

其他参考项目：


SPEED

## DIRECTION

功能：设定打印方向，此设定将被记录于 EEPROM

语法：DIRECTION n


参 数	说 明
n	0 或 1 请参考下方图例





## REFERENCE

功能：定义标签纸上的相对于原点的参考点坐标，请参考下方图例



语法：

REFERENCE x, y

参 数

x

y

说 明

水平坐标，单位“dot”

垂直坐标，单位“dot”

注意事项：**200 DPI: 1 mm = 8 dots**

**300 DPI: 1 mm = 12 dots**

范例：

REFERENCE 10,10

其他参考项目：

DIRECTION

---

## CODEPAGE

功能：该指令用于选择对应的国际字符集

语法：CODEPAGE n

<u>参</u>	<u>数</u>	<u>说</u>	<u>明</u>
n		代码页编号	

范例：

CODEPAGE 0

---

## CLS

功能：清除数据缓存

语法：CLS

注意事项：此项指令必须置于 **SIZE** 指令之后

范例：

CLS

其他参考项目：

SIZE, GAP

---

## FEED

功能：将标签纸向前推送指定的长度

语法：FEED n

参 数	说 明
n	单位: dot $1 \leq n \leq 9999$

范例：

FEED 40

注意事项：**200 DPI: 1 mm = 8 dots**  
**300 DPI: 1 mm = 12 dots**

其他参考项目：

BACKFEED, SIZE, GAP, HOME, FORMFEED

## BACKFEED&BACKUP

功能：将标签纸向后回拉指定的长度

语法：BACKFEED n

参 数	说 明
n	单位：dot $1 \leq n \leq 9999$

警告：不当的回拉距离会造成“卡纸”的现象 注意事项：  
**200 DPI: 1 mm = 8 dots**  
**300 DPI: 1 mm = 12 dots**

范例：

**BACKFEED 40**  
**BACKUP 40**

其他参考项目：

FEED, SIZE, GAP, HOME, FORMFEED

## FORMFEED

功能：将标签向前推送至下一张标签的起始位置

语法：FORMFEED

范例：

```
SIZE 50 mm,40 mm
GAP 0 mm,0 mm
SPEED 4
DENSITY 3
DIRECTION 0
OFFSET 0 mm
REFERENCE 0 mm,0 mm
SET PEEL OFF
SET COUNTER @0 +1
@0="000001"
FORMFEED
CLS
BOX 1,1,360,65,12
TEXT 25,25,"3",0,1,1,"FORMFEED COMMAND TEST"
TEXT 25,80,"3",0,1,1,@0
PRINT 3,1
```

其他参考项目：

FEED, SIZE, GAP, HOME, BACKFEED

## HOME

功能：在使用含有间隙或黑标的标签纸时，若不能确定第一张标签纸是否在正确打印位置时，此指令可将标签纸向前推送至下一张标签纸的起点开始打印。

语法：HOME

范例：

```
SIZE 50 mm, 40 mm
GAP 2 mm, 0 mm
SPEED 4
DENSITY 7
DIRECTION 0
OFFSET 0 mm
REFERENCE 0, 0
SET PEEL OFF
SET COUNTER @0 +1
@0="000001"
HOME
CLS
BOX 1, 1, 360, 65, 12
TEXT 25, 25, "3", 0, 1, 1, "HOME COMMAND TEST"
TEXT 25, 80, "3", 0, 1, 1, @0
PRINT 3, 1
```

其他参考项目：

FEED, SIZE, GAP, FORMFEED

## PRINT


功能：将存于数据缓存的标签打印

语法：PRINT m [,n]

参 数	说 明
m	打印张数 $1 \leq m \leq 999999999$
N	每张标签需重复打印的张数 $1 \leq n \leq 999999999$

范例：

```
SIZE 60 mm, 40 mm
SET COUNTER @1 1
@1="000"
CLS
TEXT 10,10,"3",0,1,1,@1
PRINT 3,2
```



其他参考项目：

SET COUNTER



## SOUND

功能：控制打印机蜂鸣器发声的频率，共有 10 阶，并可由“interval”参数设定音与音的间隔时间

语法：SOUND level,interval

参 数	说 明
level	音阶：0~9
interval	间隔时间：1~4095

范例：

```
SOUND 5,200
SOUND 3,200
SOUND 3,200
SOUND 4,200
SOUND 2,200
SOUND 2,200
SOUND 1,200
SOUND 2,200
SOUND 3,200
SOUND 4,200
SOUND 5,200
```

## SELFTEST

功能：不经自测动作，直接将打印机讯息印在标签纸上。

语法：

SELFTEST

范例：

SELFTEST

# 卷标内容设计指令

## BAR

功能:画线或画长条物

语法: BAR x,y,width,height

参 数	说 明
x	线条左上角 X 座标, 单位 dot
y	线条左上角 Y 座标, 单位 dot
width	线条宽度, 单位 dot
height	线条高度, 单位 dot

注意事项: **200 DPI: 1 mm = 8 dots**  
**300 DPI: 1 mm = 12 dots**

范例:

```

SIZE 4,2.5
GAP 0,0
SPEED 6
DENSITY 3
DIRECTION 0
CLS
BAR 100, 100, 300, 200
PRINT 1,1
    
```



其他参考项目:  
 BOX

## BARCODE

功能：印制一维条码，下方为所支持的条码列表。

- Code 128
- Code 39
- Code 93
- EAN 13
- EAN 8
- Coda bar
- UPC-A
- UPC-E
- ITF

语法：

BARCODE X, Y, "code type", height, human readable, rotation, narrow, wide, "code"

参 数	说 明
X	明定条码左上角 X 座标
Y	明定条码左上角 Y 座标
条码类型	
height	条码高度 (dot)
human readable	0：无人眼可识别码 1：有人眼可识别码
rotation	顺时针旋转条码角度
0	不旋转
90	顺时针旋转 90 度
180	顺时针旋转 180 度
270	顺时针旋转 270 度
narrow	窄条码比例因子 (dot)
wide	宽条码比例因子 (dot)

范例：

BARCODE 100,100,"39",96,1,0,2,4,"1000"  
 BARCODE 10,10,"128",48,1,0,2,2,"!123456799!"


## BITMAP

功能：绘制 BITMAP 格式的图形 (非 BMP 格式的图档)

语法：BITMAP X, Y, width, height, mode, bitmap data...

参 数	说 明
X	明定图形左上角 X 座标
Y	明定图形左上角 Y 座标
width	图形的宽度，单位为 byte
height	图形的高度，单位为 dot
mode	绘制图形的方式
0	OVERWRITE
1	OR
2	XOR
bitmap data	图形资料

范例：



ROW (Y-axis)	L-		R-	
	Binary	Hexadecimal	Binary	Hexadecimal
0	00000000	0	00000000	0
1	00000000	0	00000000	0
2	00000000	0	00000000	0
3	00000111	0	11111111	F
	00000011	0	11111111	F
5	00010001	1	11111111	F
6	00011000	1	11111111	F
7	00011100	1	01111111	7
8	00011110	1	00111111	3
9	00011111	1	00011111	1
A	00011111	1	10001111	8
B	00011111	1	11000111	C
C	00011111	1	11100011	E
D	00011111	1	11110111	F
E	00011111	1	11111111	F
F	00011111	1	11111111	F

范例:

SIZE 4,2

GAP 0,0

CLS

BITMAP 200,200,2,16,0,

-?????

PRINT 1,1

Hexadecim	AS
53 49 5A 45 20 34 2C 32 0D 0A 47 41 50	SIZE 4,2
20	GAP 0,0
30 2C 30 0D 0A 43 4C 53 0D 0A 42 49 54	CLS
4D	BITMAP
41 50 20 32 30 30 2C 32 30 30 2C 32 2C	200,200,2,16,0,
31	
36 2C 30 2C 00 00 00 00 00 07 FF 03	?-
FF	????
11 FF 18 FF 1C 7F 1E 3F 1F 1F 8F 1F	PRINT 1,1
C7	

其他参考项目:

PUTBMP, PUTPCX

## BOX

功能：绘制方框

语法：BOX X\_start, Y\_start, X\_end, Y\_end, line thickness

参 数	说 明
X_start	方框左上角 X 座标，单位：dot
Y_start	方框左上角 Y 座标，单位：dot
X_end	方框右下角 X 座标，单位：dot
Y_end	方框右下角 Y 座标，单位：dot
line thickness	方框线条粗细，单位：dot

注意事项：**200 DPI: 1 mm = 8 dots**  
**300 DPI: 1 mm = 12 dots**

范例：

SIZE 60 mm,40 mm  
GAP 0,0  
SPEED 6  
DENSITY 3  
DIRECTION 0  
CLS  
BOX 100,100,200,200,5  
PRINT 1,1

(100,100)



(200,200)

其他参考项目：

BAR

## ERASE

功能：清除影像缓冲区中指定大小的区域


语法：ERASE X\_start, Y\_start, X\_width, Y\_height

参 数	说 明
X_start	欲清除区域的左上角 X 座标，单位：dot
Y_start	欲清除区域的左上角 Y 座标，单位：dot
X_width	欲清除区域的宽度，单位：dot
Y_height	欲清除区域的高度，单位：dot

范例：

```

SIZE 60 mm,60 mm
GAP 0,0
SPEED 6
DENSITY 3
DIRECTION 0
CLS
BAR 100, 100, 300, 300
ERASE 150,150,200,200
PRINT 1,1
    
```



其他参考项目：  
CLS



## PUTBMP

功能：打印 BMP 格式图档

语法：PUTBMP X, Y, "filename"

<u>参 数</u>	<u>说 明</u>
X	BMP 图形左上角 X 座标
Y	BMP 图形左上角 Y 座标
filename	已载至打印机的 BMP 图档

注意事项：建议使用仅有黑白二种颜色的 **BMP** 图档

其他参考项目：

DOWNLOAD, BITMAP, PUTPCX

## PUTPCX

功能：打印 PCX 格式图档 语法：PUTPCX X, Y, " filename"

<u>参 数</u>	<u>说 明</u>
X	PCX 图形左上角 X 座标
Y	PCX 图形左上角 Y 座标
filename	已载至打印机的 PCX 图档

其他参考项目：

DOWNLOAD, BITMAP, PUTPCX

## QRCODE

功能：绘制 QRCODE 二维条码

语法：

QRCODE X, Y, ECC Level, cell width, mode, rotation, [model, mask,]"Data string"

参 数	说 明
X	QRCODE 条码左上角 X 座标
Y	QRCODE 条码左上角 Y 座标
ECC level	错误纠正能力等级
L	7%
M	15%
Q	25%
H	30%
cell width	1~10
mode	自动生成编码/手动生成编码
A	Auto
M	Manual
rotation	顺时针旋转角度
0	不旋转
90	顺时针旋转 90 度
180	顺时针旋转 180 度
270	顺时针旋转 270 度
model	条码生成样式
1	(预设), 原始版本
2	扩大版本
mask	范围：0~8, 预设 7
Data string	条码资料内容

可用编码字符集：

- 1). 数值资料：数字 0~9
- 2). 文字数字资料：数字 0~9；大写字母 A-Z；其他：space, \$%\*+-./,GB18030字符集:

试列：

SIZE 60 mm,60 mm

CAP 0,0

CLS

QRCODE 10,10,H,4,M,0,"AABC!B0005\["]abc\["!]N123"

QRCODE 310,310,H,4,M,0,"B0001\["!]K 打印机!B0010\["]ABCabc123"

PRINT 1,1

## REVERSE

功能：将图像缓存中指定的区域做反相的动作

语法：REVERSE X\_start, Y\_start, X\_width, Y\_height

参 数	说 明
X_start	欲清除区域的左上角 X 座标，单位：dot
Y_start	欲清除区域的左上角 Y 座标，单位：dot
X_width	欲清除区域的宽度，单位：dot
Y_height	欲清除区域的高度，单位：dot

注意事项：**200 DPI: 1 mm = 8 dots**  
**300 DPI: 1 mm = 12 dots**

范例：

```
SIZE 4,2.5  
GAP 0,0  
SPEED 6  
DENSITY 3  
DIRECTION 0  
CLS  
TEXT 100,100,"3",0,1,1,"REVERSE"  
REVERSE 90,90,128,40  
PRINT 1,1
```



## TEXT

功能：打印文字

语法：TEXT X, Y, "font", rotation, x-multiplication, y-multiplication, "content"

参 数	说 明
X	文字方块左上角 X 座标
Y	文字方块左上角 Y 座标
font	字型名称
1	8 x 12 英数字体
2	12 x 20 英数字体
3	16 x 24 英数字体
4	24 x 32 英数字体
5	32 x 48 英数字体
6	14 x 19 英数字体 OCR-B
7	14 x 25 英数字体 OCR-A
8	21 x 27 英数字体 OCR-B
9	9 x 17 英数字体
TST24.BF2	繁体中文 24x24 字体(大五码)
TSS24.BF2	简体中文 24x24 字体(GB 码)
Rotation	顺时针旋转角度
0	不旋转
90	顺时针旋转 90 度
180	顺时针旋转 180 度
270	顺时针旋转 270 度
X-multiplication:	水平放大值，最大可放大至 10 倍 有效系数：1~10
Y-multiplication:	垂直放大值，最大可放大至 10 倍 有效系数：1~10

注意事项：若文字内容包括双引号(")时，则请用[V"] 取代  
 若要打印 **0D(hex)** 字符时，请在程序中使用[V]R]来打印 **CR**  
 若要打印 **0A(hex)** 字符时，请在程序中使用[V]A]来打印 **LF**  
 五号字英文字母仅可打印大写字母

范例 1:

SIZE 72 mm,60 mm

CLS

TEXT 10,10,"1",0,1,1,"AB0CDEFGHIJKLMNOPQRSTUVWXYZ"

TEXT 10,30,"2",0,1,1,"AB0CDEFGHIJKLMNOPQRSTUVWXYZ"

TEXT 10,60,"3",0,1,1,"AB012CDEF"

---

TEXT 10,90,"4",0,1,1,"AB012CDEF456UVWXYZ"

TEXT 10,130,"5",0,1,1,"AB0CDEFGHIJKLMNOPQRSTUVWXYZ"

TEXT 10,190,"6",0,1,1,"AB0CDEFGHIJKLMNOPQRSTUVWXYZ"

TEXT 10,220,"7",0,1,1,"AB0CDEFGHIJKLMNOPQRSTUVWXYZ"

TEXT 10,250,"8",0,1,1,"AB0CDEFGHIJKLMNOPQRSTUVWXYZ"

TEXT 10,280,"0",0,1,1,"AB0CDEFGHIJKLMNOPQRSTUVWXYZ"

TEXT 10,310,"9",0,1,1,"AB0CDEFGHIJKLMNOPQRSTUVWXYZ"

TEXT 10,330,"TSS24.BF2",0,1,1,"欢迎使用资江电子标签打印机"

PRINT 1,1

## 询问打印机状态指令

### <ESC>!?

功能: 透过串口送出指令用来获得打印机目前状态, 其中<ESC>脱离符号表示 ASCII 27 (Hex1B), 此项指令可随时被送出, 即使是在打印机发生错误状态时。

<u>位</u>	<u>状 态</u>
0	就绪
4	缺纸

语法: <ESC>!?

其他参考项目:  
<ESC>!R

## ~!T

功能：查询打印机的型号，经 RS-232 以 ASCII 字符回传

语法：~!T

范例：

~!T

其他参考项目：

~!I, ~!F

# 文件管理指令

## DOWNLOAD

功能：“DOWNLOAD”为一文件内容的首行，定义文件可被储存于打印机的 DRAM 中。下载至

打印机的文件可区分为两种：程序文件及资料文件（包括字符型文件、PCX 图形文件和 BMP 图形文件等）

指令语法：

1. 下载程序文件：

DOWNLOAD [n,]“FILENAME.BAS”

参 数

n

说 明

指定储存文件的位置

不指定时：文件被储存于 DRAM

F：文件被储存于 FLASH

FILENAME.BAS

储存于打印机中的文件名称

注意事项：

(1). 大小写将表示不同的文件名称

(2). 程序文件的扩展名必须为“.BAS”

(3). 若不指定储存的位置，则文件一律载至 **DRAM**，存于 **DRAM** 的文件会

因电源关闭而消失

2. 下载资料文件：

DOWNLOAD [n,]“FILENAME”, DATA SIZE, DATA CONTENT...

参 数

n

说 明

指定储存文件的位置

不指定时：文件被储存于 DRAM.

F：文件被储存于 FLASH

FILENAME

储存于打印机中的文件名称

DATA SIZE

不含文件头的实际文件大小，以字节数计算

注意事项：

(1). 行与行之间的资料以 **CR(0x0D)**及 **LF(0x0A)**做分隔

(2). 若不指定储存的位置，则文件一律载至 **DRAM**，存于 **DRAM** 的文件会

因电源关闭而消失



范例:

下方程序范例会将文件载至打印机的 DRAM

```
DOWNLOAD  
"EXAMPLE.BAS" SIZE 4,4  
GAP 0,0  
DENSITY 8  
SPEED 6  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL  
OFF CLS  
TEXT 100,100,"3",0,1,1,"EXAMPLE  
PROGRAM" PRINT 1  
EOP
```

注意事项: 当编写一个可被下载至打印机的程序时, "**DOWNLOAD**"必须置于文件的首行, "**EOP**"必须置于程序的最后一行。执行程序的方式有二: 一为调用主档名(不含 **BAS**)执行, 另一为使用 **RUN** 指令执行程序。

范例:

1. 调用主档名执行:

```
C:\>COPY CON LPT1<ENTER>  
EXAMPLE<ENTER>  
<CTRL><Z>  
C:\>
```

2. 使用 **RUN** 指令执行程序:

```
C:\>COPY CON LPT1<ENTER>  
RUN "EXAMPLE.BAS"<ENTER>  
<CTRL><Z>  
C:\>
```

下方为下载资料文件的范例.

```
DOWNLOAD "DATA",20,COMPUTER<Enter>  
2001<Enter>  
21<Enter>
```

注意事项: 上方范例中 **<ENTER>**表示按下键盘上的"**ENTER**"键。

其他参考项目:

EOP, RUN, PUTBMP, PUTPCX

---

## EOP

功能：作为载入程序文件的结尾。当使用 BASIC 语法时，DOWNLOAD “FILENAME”必须置于文件首行，EOP 必须置于文件结尾。

语法：

EOP

范例：

```
DOWNLOAD "DEMO.BAS"  
SIZE 4,4  
GAP 0,0  
DENSITY 8  
SPEED 6  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL OFF  
CLS  
TEXT 100,100,"3",0,1,1,"DEMO PROGRAM"  
PRINT 1  
EOP
```

其他参考项目：

DOWNLOAD, EOP

## FILES

功能：此指令会将已载入打印机存储器的文件名称打打印来

语法：

FILES

范例：

下列步骤为在 DOS 模式下透过并口或串口将打印机内的文件打打印来

串口：

```
C:\>MODE COM1 96,N,8,1<ENTER>
```

```
C:\>COPY CON COM1<ENTER>
```

```
FILES<ENTER>
```

```
<CTRL><Z><ENTER>
```

```
C:\>
```

并口：

```
C:\>COPY CON LPT1<ENTER>
```

```
FILES<ENTER>
```

```
<CTRL><Z><ENTER>
```

注意事项：上方范例中 **<ENTER>** 表示按下键盘上的 **"ENTER"** 键， **<CTRL>** 表示按下键盘上的 **"Ctrl"** 键， **<Z>** 表示按下键盘上的 **"Z"** 键。

其他参考项目：

~!F, KILL

# 打印机外围功能设定指令

## SET COUNTER

功能: COUNTER 可以是一般的计数器或是一变量。可设定计数器的初始值及计数使用的增量。

语法:

```
SET COUNTER @n step
@n = "Expression"
```

参 数

@

step

Expression

说 明

n: 计数器的初始值。总计有 50(@0~@49)个计数器可同时使用

计数器的增量值可为正数亦可为负数。

-999999999<= step <=999999999

如果计数器只是被当做固定的变量使用时, 请将增量值设为 0 初始字符串

范例:

```
SIZE 3,3
GAP 0,0
DENSITY 8
SPEED 6
DIRECTION 0
REFERENCE 0,0
SET COUNTER @1 1
@1="00001"
SET COUNTER @2 5
@2="AB000001"
CLS
TEXT 50,50,"3",0,1,1,@1
BARCODE 50,100,"39",48,1,0,2,4,@2
PRINT 2,1
```

其他参考项目:

PRINT, TEXT, BARCODE

## SET PEEL

功能：设定启动/关闭自动剥纸器功能。预设值为关闭状态，当此功能被开启时，打印机会在每印完一张时即暂停，直到标签纸被取走后才会打印下一张标签。此项设定会被记录于打印机中。

语法：

SET PEEL ON/OFF

参 数	说 明
ON	开启自动剥纸器的功能
OFF	关闭自动剥纸器的功能

范例：

```
REM ***SELF-PEELING FUNCTION ON***  
SIZE 4,4  
GAP 0,0  
DENSITY 8  
SPEED 6  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL ON  
CLS  
TEXT 50,100,"3",0,1,1,"SELF-PEELING FUNCTION TEST"  
PRINT 5
```

其他参考项目：

OFFEST, PRINT

## SET TEAR & SET STRIPPER

功能：设定开启/关闭送纸至撕纸线的功能，此项设定会被记录于打印机中

语法：

SET TEAR ON/OFF

参 数

ON

OFF

说 明

标签打印结束时将送纸至撕纸位置

标签打印结束时会将标签起印点停留至打印线位置

范例：

```
REM ***TEAR FUNCTION ON***  
SIZE 3,3  
GAP 0.08,0  
DENSITY 8  
SPEED 4  
DIRECTION 0  
REFERENCE 0,0  
SET PEEL OFF  
SET TEAR ON  
CLS  
TEXT 50,100,"3",0,1,1,"TEAR FUNCTION TEST"  
PRINT 1
```

其他参考项目：

SET PEEL

---

## BEEP

功能：打印机收到该指令会响一声

语法：BEEP

范例：

SIZE 60 mm,40 mm

GAP 0,0

DENSITY 8

SPEED 6

DIRECTION 0

REFERENCE 0,0

SET PEEL OFF

CLS

BEEP

TEXT 100,100,"0",0,1,1,"资江电子123"

PRINT 1,1